


الباب الثالث  
ماتيماتيكا والجبر



في هذا الباب سوف نتعرف على أوامر برنامج ماتيماتيكا  
والخاصة بالموضوعات الآتية :

١ . كثيرات الحدود والدوال الكسرية Polynomials & Rational Functions

٢ . المتسلسلات Series

٣ . حل المعادلات Solving Equations

٤ . الجبر الخطي Linear Algebra

أولا : القوائم Lists

ثانيا : المصفوفات Matrices

ثالثا : حل الأنظمة الخطية Solving Linear Systems

رابعا : القيم المميزة والمتجهات المميزة Eigenvalues and Eigenvectors



## الباب الثالث

### ماثياتيكا والجبر

من الأشياء الهامة في برنامج ماثياتيكا هو المقدرة على القيام بالحسابات على المقادير الرمزية **Symbolic** الى جانب المقادير العددية **Numeric** وهذا يعني أن ماثياتيكا يستطيع التعامل مع الصيغ والمقادير الجبرية تماما مثل التعامل مع الأعداد .

#### ١ . كثيرات الحدود والدوال الكسرية Polynomials and Rational Functions

برنامج ماثياتيكا يقدم عدد كبير من الأوامر للتحويل بين الأشكال المختلفة للتعبيرات الجبرية وأجراء العمليات الجبرية على كثيرات الحدود **Polynomials** وسوف نتعرف فيما يأتي على بعض هذه الأوامر مع توضيح الوظيفة والصيغة العامة لكل من هذه الأوامر ونعطى أمثلة توضيحية .

الصيغة العامة للأمر	العمل الذي يقوم به الأمر
<b>Factor[poly]</b>	تحليل كثيرة الحدود <b>Poly</b> الى قوى صحيحة
<b>Expand[expr]</b>	إيجاد مفكوك حاصل الضرب والقوى الصحيحة الموجبة الموجودة في البسط للتعبير <b>expr</b>
<b>ExpandAll[expr]</b>	إيجاد مفكوك حاصل الضرب والقوى الصحيحة الموجبة الموجودة في كل أجزاء التعبير <b>expr</b>
<b>Together[expr]</b>	توحيد المقامات <b>denominators</b> للكسور الموجودة في التعبير <b>expr</b>
<b>Apart[expr]</b>	كتابة التعبير الكسرى <b>expr</b> على صورة مجموع لكسوره الجزئية
<b>Simplify[expr]</b>	إيجاد صورة مبسطة للتعبير <b>expr</b> بأصغر عدد ممكن من الأجزاء
<b>Collect[expr,x]</b>	تجميع الحدود التي تحتوي على نفس قوى <b>x</b> في التعبير <b>expr</b>

In[1]:=Factor[x^2+2x-3]

لتحليل كثيرة الحدود  $x^2 + 2x - 3$

Out[1]= (x+3)(x-1)

إلى عواملها

In[2]:=Expand[%]

الامر Expand هو عكس الامر Factor

Out[2]=  $x^2 + 2x - 3$

ويقوم بفك الأقواس

In[3]:=Factor[8x^3+36x^2+54x+27]

لتحليل كثيرة الحدود

Out[3]=  $(3 + 2x)^3$

$8x^3 + 36x^2 + 54x + 27$

إلى عواملها

تعريف المقدار الجبري rrr ثم إيجاد مفكوك القوى الصحيحة الموجبة وكذلك حاصل

الضرب الموجود في البسط للمقدار الجبري rrr

In[4]:=rrr=(x-1)^2 (x+2)/((x+1)(x-3)^2);Expand[rrr]

$$\text{Out[4]} = \frac{2}{(-3+x)^2(1+x)} - \frac{3x}{(-3+x)^2(1+x)} + \frac{x^3}{(-3+x)^2(1+x)}$$

In[5]:=Together[%]

توحيد المقامات للكسور الموجودة في المقدار

$$\text{Out[5]: } \frac{2 - 3x + x^3}{(-3 + x)^2(1 + x)}$$

الجبري الناتج من جملة الإخراج السابقة

إيجاد مفكوك القوى الصحيحة الموجبة وكذلك حاصل الضرب الموجود في كل المقدار الجبري rrr

In[6]:=ExpandAll[rrr]

$$\text{Out[6]= } \frac{2}{9+3x-5x^2+x^3} - \frac{3x}{9+3x-5x^2+x^3} + \frac{x^3}{9+3x-5x^2+x^3}$$

In[7]:=Apart[rrr]

كتابة المقدار الجبري rrr في صورة

$$\text{Out[7]= } 1 + \frac{5}{(-3+x)^2} + \frac{19}{4(-3+x)} + \frac{1}{4(1+x)}$$

كسوره الجزئية

In[8]:=Simplify[%7]

تبسيط شكل الناتج من جملة الإخراج

$$\text{Out[8]= } \frac{(-1 + x)^2(2 + x)}{9 + 3x - 5x^2 + x^3}$$

السابقة Out[7]

الصيغة العامة للأمر	العمل الذى يقوم به الأمر
<code>Collect[expr, x]</code>	تجميع الحدود التى تحتوى على نفس قوى $x$ فى التعبير $expr$
<code>Coefficient[expr, form]</code>	الحصول على معامل $form$ فى كثيرة الحدود $expr$
<code>Length[expr]</code>	الحصول على عدد العناصر الموجودة فى التعبير $expr$
<code>Exponent[expr, form]</code>	الحصول على أكبر قوى للمقدار $form$ فى التعبير $expr$

`In[9]:=r1=Expand[(2x+y+1)^2]` إيجاد مفكوك المقدار الجبرى  $(2x + y + 1)^2$

`Out[9]= 1 + 4 x + 4 x^2 + 2 y + 4 x y + y^2` ووضع الناتج فى المتغير  $r1$

`In[10]:=Collect[r1,y]` تجميع الحدود التى تحتوى على نفس قوى المتغير  $y$  فى التعبير الجبرى  $r1$

`Out[10]= 1 + 4 x + 4 x^2 + (2 + 4 x) y + y^2`

`In[11]:=Coefficient[r1,y]` للحصول على معامل  $y$  فى التعبير الجبرى  $r1$

`Out[11]=2 + 4 x`

`In[12]:=Exponent[r1,y]` للحصول على أكبر قوى للمتغير  $y$  فى التعبير الجبرى  $r1$

`Out[12]=2`

الصيغة العامة للأمر	العمل الذي يقوم به الأمر
Numerator[expr]	الحصول على البسط في التعبير expr
Denominator[expr]	الحصول على المقام في التعبير expr
PolynomialQuotient[p, q, x]	إيجاد خارج قسمة p على q مع إهمال الجزء الباقي حيث p, q كثيرات حدود في المتغير x
PolynomialRemainder[p, q, x]	إيجاد الجزء الباقي من خارج قسمة p على q حيث p, q كثيرات حدود في المتغير x

In[13]:=Numerator[rrr]

للحصول على البسط في التعبير الجبري rrr

Out[13]=(-1+x)<sup>2</sup> (2+x)

In[14]:=Denominator[rrr]

للحصول على المقام في التعبير الجبري rrr

Out[14]=(-3+x)<sup>2</sup> (1+x)

In[15]:= p=x<sup>3</sup>+5x<sup>2</sup>+4x-6;q=x+1;

تعريف كثيرتي حدود p,q ثم

PolynomialQuotient[p,q,x]

إيجاد خارج قسمة كثيرة الحدود p على

Out[15]=4x<sup>2</sup> +x

كثيرة الحدود q مع إهمال الجزء الباقي

In[16]:=PolynomialRemainder[p,q,x]

للحصول على الجزء الباقي من خارج قسمة

Out[16]= -6

كثيرة الحدود p على كثيرة الحدود q

## Series . المتسلسلات

في كثير من المسائل الرياضية تنشأ عمليات جمع وضرب للحدود المنتظمة وبرنامج ماتيماتكا قادر على حساب مثل هذه العمليات . وحساب مجموع حدود المتسلسلة يستخدم الأمر Sum كالآتي :

الصيغة العامة للأمر	العمل الذي يقوم به الأمر
Sum[f, {i, imax}]	حساب المجموع $\sum_{i=1}^{i \max} f$
Sum[f, {i, imin, imax}]	حساب المجموع $\sum_{i=i \min}^{i \max} f$
Sum[f, {i, imin, imax, step}]	حساب المجموع $\sum_i f$ من $i=i \min$ الى $i=i \max$ بخطوة مقدارها step
Sum[f, {i, imin, imax}, {j, jmin, jmax}, ...]	حساب المجموع $\sum_{i=i \min}^{i \max} \sum_{j=j \min}^{j \max} f$

In[1]:= Sum[1/i^2,{i,1,10}]/N

Out[1]=1.54977

حساب مجموع العشرة حدود الأولى

من المتسلسلة  $\sum_{i=1}^{10} \frac{1}{i^2}$

In[2]:= Sum[i/I^2,{i,1,Infinity}]/N

ويمكن حساب مجموع عدد لانهاى من

Out[2]= 1.64493

المتسلسلة بشرط أن تكون المتسلسلة تقاربية

In[3]:=Sum[x^i/i!,{i,1,7,2}

حساب المجموع  $\sum_{i=1}^7 \frac{x^i}{i!}$

$$\text{Out}[3]= x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + \frac{x^5}{120} + \frac{x^6}{720} + \frac{x^7}{5040}$$

In[4]:=Sum[x^i/i!,{i,1,7,2}]

حساب مجموع  $\sum_{i=2}^7 \frac{x^i}{i!}$  من  $i=2$  الى  $i=7$

$$\text{Out}[4]= x + \frac{x^3}{6} + \frac{x^5}{120} + \frac{x^7}{5040}$$

بخطوة 2=step

In[5]:=Sum[x^i y^j,{i,1,3},{j,1,i]}

حساب المجموع المزدوج  $\sum_{i=1}^3 \sum_{j=1}^i x^i y^j$

$$\text{Out}[5]= xy + x^2y + x^2y^2 + x^3y + x^3y^2 + x^3y^3$$

ويمكن لماتيماتيكا إجراء عمليات الضرب على الحدود المنتظمة باستخدام الأمر **Product** كالآتي :

الصيغة العامة للأمر	العمل الذي يقوم به الأمر
<b>Product[f,{i, imax}]</b>	حساب حاصل الضرب $\prod_{i=1}^{imax} f$
<b>Product[f,{i, imin, imax}]</b>	حساب حاصل الضرب $\prod_{i=imin}^{imax} f$
<b>Product[f,{i, imin, imax, step}]</b>	حساب حاصل الضرب $\prod_{i=imin}^{imax} f$ بخطوة مقدارها <b>step</b>
<b>Product[f,{i,imin, imax}, {j, jmin, jmax}, ...]</b>	حساب حاصل الضرب المزدوج $\prod_{i=imin}^{imax} \prod_{j=jmin}^{jmax} f$

**In[6]:=Product[i^2,{i,1,5}]**

حساب حاصل الضرب  $\prod_{i=1}^5 i^2$

**Out[6]=14400**

**In[7]:=Product[x+i,{i,1,4}]**

حساب حاصل الضرب  $\prod_{i=1}^4 (x+i)$

**Out[7]=(1 + x) (2 + x) (3 + x) (4 + x)**

**In[8]:=Product[(x+i)^j,{i,1,3},{j,1,i}]**

حساب حاصل  $\prod_{i=1}^3 \prod_{j=1}^i (x+i)^j$

**Out[8]=(1 + x) (2 + x)<sup>3</sup> (3 + x)<sup>6</sup>**

وبرنامج ماتيماتيكا قادر على حساب مفكوك متسلسلة القوى للدالة  $f(x)$  حول النقطة  $x = x_0$  لأي عدد  $n$  من الحدود وكذلك حساب مفكوك تيلور لدالة في متغيرين  $f(x,y)$  حول النقطة  $(x_0, y_0)$  وذلك باستخدام الأمر Series كالتالي :

الصيغة العامة للأمر	العمل الذي يقوم به الأمر
<code>Series[f, {x, x0, n}]</code>	حساب مفكوك متسلسلة القوى للدالة $f$ حول النقطة $x_0$ حتى الحد $(x-x_0)^n$
<code>Series[f, {x, x0, nx}, {y, y0, ny}]</code>	حساب مفكوك متسلسلة القوى للدالة $f$ على التابع بالنسبة الى $y$ ثم الى $x$
<code>Normal[expr]</code>	تحويل <code>expr</code> الى الشكل العادي بدون أي رموز خاصة

لحساب مفكوك متسلسلة القوى للدالة  $f(x)$  حول النقطة  $x=a$  حتى الحدود من الدرجة الثالثة

`In[9]:=Series[f[x],{x,a,3}]`

`Out[9]=`

$$f[a]+f'[a](-a+x)+\frac{f''[a](-a+x)^2}{2}+\frac{f^{(3)}[a](-a+x)^3}{6}+O[-a+x]^4$$

لحساب مفكوك متسلسلة القوى للدالة  $e^x$  حول النقطة  $x = 0$  حتى الحدود من الدرجة الرابعة

In[10]:=Series[Exp[x],{x,0,4}]

$$\text{Out}[10]=1+x+\frac{x^2}{2}+\frac{x^3}{6}+\frac{x^4}{24}+O[x]^5$$

ويمكن إلغاء الحد  $O[x]^5$  وكتابة المفكوك في الشكل العادى وذلك باستخدام الأمر Normal

In[11]:=Normal[%]

$$\text{Out}[11]=1+x+\frac{x^2}{2}+\frac{x^3}{6}+\frac{x^4}{24}$$

لحساب مفكوك متسلسلة القوى للدالة  $e^{xy}$  حول النقطة  $x = 0$  ,  $y = 0$  حتى الحدود من الدرجة الثالثة فى  $x$  والدرجة الثانية فى  $y$

In[12]:=Series[Exp[x y],{x,0,3},{y,0,2}]

$$\text{Out}[12]=1+(y+O[y]^3)x+(\frac{y^2}{2}+O[y]^3)x^2+O[x]^3$$

ويمكن كتابة المفكوك في الشكل العادى وذلك باستخدام الأمر Normal

In[13]:=Normal[Series[Exp[x y],{x,0,3},{y,0,2}]]

$$\text{Out}[13]=1+xy+\frac{x^2y^2}{2}$$

## ٣ . حل المعادلات Solving Equations

في برنامج ماتيماتيكا المعادلة يكون لها الشكل  $lhs == rhs$  حيث يستخدم المؤثر العلاقي  $==$  وهو يعنى اختبار ما إذا كان الطرف الأيمن  $rhs$  يساوى الطرف الأيسر  $lhs$  ولذلك فإن المعادلات في ماتيماتيكا تعامل على أنها تعبيرات منطقية **Logical Statement** .

فمثلا عند إدخال المعادلة  $2+3==5$  فإن الناتج يكون صواب **True**

**In[1] := 2+3==5**

**Out[1]= True**

وعند إدخال المعادلة  $x^2 + 3x == 2$  فإن ماتيماتيكا يقوم بإخراج المعادلة في صورة رمزية لأنه لم يستطع اختبار ما إذا كان  $x^2 + 3x == 2$  صواب أو خطأ نظرا لعدم وجود قيمة سابقة للمتغير  $x$  .

**In[2] := x^2+3x==2**

**Out[2]= x^2+3x==2**

والآن لحل المعادلة والحصول على قيم  $x$  التي تمثل جذور المعادلة نستخدم الأمر **Roots** كالاتي :

**Roots[lhs==rhs,var]**

للحصول على قائمة تحتوي على جذور المعادلة

. **var** بالنسبة للمتغير

الحصول على جذور المعادلة  $x^2+3x=2$  ونلاحظ أن الناتج يكون على صورة تعبير منطقي

In[3]:=Roots[x^2+3x==2,x]

Out[3]=

$$x = \frac{-3 - \text{Sqrt}[17]}{2} \quad || \quad x = \frac{-3 + \text{Sqrt}[17]}{2}$$

وكثيرا ما نحتاج الى استخدام الحل الناتج في حسابات اخرى لذلك يكون من المفيد تحويل الحل من التعبير المنطقي الى صورة صريحة ويتم ذلك باستخدام الدالة ToRules

**ToRules[eqns]** لتحويل حل eqns الناتج من الأمر Roots من الصورة المنطقية الى متابعة من القوائم تحتوي على قواعد صريحة للحل

**{ToRules[eqns]}** لتحويل حل eqns الناتج من الأمر Roots من الصورة المنطقية الى قائمة تحتوي على قواعد صريحة للحل

In[4]:=ToRules[%3]

لتحويل الحل الناتج من جملة 3

Out[4]=

الى متابعة تحتوي على قواعد صريحة للحل

$$\text{Sequence}\left\{\left\{x \rightarrow \frac{-3 - \text{Sqrt}[17]}{2}\right\}, \left\{x \rightarrow \frac{-3 + \text{Sqrt}[17]}{2}\right\}\right\}$$

In[5]:={ToRules[Roots[x^2+3x==2,x]]//N لتحويل الكسور الاعتيادية في الحل

Out[5]={{x -> -3.56155}, {x -> 0.561553}} الى كسور عشرية والحصول على قيم

عددية نستخدم الدالة N

ويمكن حل المعادلة والحصول على جذورها باستخدام الأمر Solve كالاتي :

الحل المعادلة eqn بالنسبة للمتغير var Solve[eqn, var]

وتعتبر معادلات كثيرات الحدود polynomial equations من أهم المعادلات التي يتم حلها باستخدام الأمر Solve .

للحصول على جذور المعادلة  $x^2 + 3x = 2$  بالنسبة للمتغير x

In[6]:=Solve[x^2+3x=-2,x]

Out[6]= {x ->  $\frac{-3 - \text{Sqrt}[17]}{2}$ }, {x ->  $\frac{-3 + \text{Sqrt}[17]}{2}$ }

لحساب قيمة عددية تقريبية للحل الناتج من جملة الإدخال رقم 6

In[7]:=N[%6]

Out[7]={{x -> -3.56155}, {x -> 0.561553}}

للحصول على جذور المعادلة  $a x + b = c$  بالنسبة للمتغير x

In[8]:=Solve[a x+b=-c,x]

Out[8]= {{x ->  $-\left(\frac{b-c}{a}\right)}$ }

للحصول على جذور المعادلة  $ax^2 + bx + c = 0$  بالنسبة الى المتغير  $x$

In[9]:=Solve[a x^2+b x+c==0,x]

$$\text{Out[9]} = \left\{ \left\{ x \rightarrow \frac{-b - \text{Sqrt}[b^2 - 4ac]}{2a} \right\}, \left\{ x \rightarrow \frac{-b + \text{Sqrt}[b^2 - 4ac]}{2a} \right\} \right\}$$

للحصول على جذور المعادلة  $x^3 + 3x^2 + 3x + 2 = 0$  بالنسبة الى المتغير  $x$

In[10]:=Solve[x^3+3x^2+3x+2==0,x]

$$\text{Out[10]} = \left\{ \{x \rightarrow -2\}, \left\{ x \rightarrow \frac{-1 - \text{I Sqrt}[3]}{2} \right\}, \left\{ x \rightarrow \frac{-1 + \text{I Sqrt}[3]}{2} \right\} \right\}$$

ولحساب قيمة عددية تقريبية للحل الناتج من جملة الإدخال رقم 10

In[11]:=N[%]

Out[11]={{x -> -2.}, {x -> -0.5 - 0.866025 I}, {x -> -0.5 + 0.866025 I}}

ويمكن الحصول على جذور معينة من حل المعادلة باستخدام الأقواس المزدوجة [ ]

In[12]:=Solve[x^3+3x^2+3x+2==0,x][[1]] للحصول على الجذر الأول من حل

Out[12]={x -> -2} المعادلة  $x^3 + 3x^2 + 3x + 2 = 0$

In[13]:=Solve[x^3+3x^2+3x+2==0,x][[2]]/N للحصول على قيمة عددية للجذر

Out[13]={x -> -0.5 - 0.866025 I} الثاني من حل المعادلة  $x^3 + 3x^2 + 3x + 2 = 0$

والأمر **Solve** قادر على إيجاد حلول جبرية صريحة للعديد من معادلات كثيرات الحدود ذات الدرجات العالية خاصة المعادلات التي يمكن تحليلها

In[14]:=eq1=Expand[Product[x-i,{i,5}]] تعريف كثيرة حدود من الدرجة الخامسة  
 Out[14]=-120 + 274 x - 225 x<sup>2</sup> + 85 x<sup>3</sup> - 15 x<sup>4</sup> + x<sup>5</sup>

In[15]:=Solve[eq1==0,x] حل معادلة كثيرة حدود من الدرجة الخامسة  
 Out[15]={{x -> 1}, {x -> 2}, {x -> 3}, {x -> 4}, {x -> 5}}

نلاحظ أننا حصلنا على حل صريح لمعادلة كثيرة الحدود eq1=0 من الدرجة الخامسة .  
 وإذا كان ماتيماتيكا قادر على إيجاد حلول معادلة كثيرة حدود من درجة n فإنه يعطى n من الجذور حتى في حالة وجود جذور مكررة كما في المثال الآتي :

In[16]:=Solve[(x+3)(x-1)^2==0,x]  
 Out[16]={{x -> -3}, {x -> 1}, {x -> 1}}

وفي حالة عدم استطاعة ماتيماتيكا الحصول على حلول جبرية صريحة فإن ماتيماتيكا تترك المعادلة في صورتها الرمزية ويمكن في هذه الحالة استخدام الدالة N للحصول على حلول عديدة

In[17]:=Solve[x^5-130x+11==0,x]  
 Out[17]={ToRules[Roots[-130 x + x<sup>5</sup> == -11, x]]}

In[18]:= Solve[x^5-130x+11==0,x]/N  
 Out[18]={{x -> -3.39748}, {x -> -0.0211456 - 3.37698 I},  
 {x -> -0.0211456 + 3.37698 I}, {x -> 0.0846154}, {x -> 3.35515}}

وفي ماتيماتيكا يمكن استخدام الأمر Solve لحل بعض المعادلات التي ليست على صورة كثيرات حدود

In[19]:=Solve[Sqrt[1-x]+Sqrt[1+x]=4,x]/N  
Out[19]={{x -> -6.9282 I}, {x -> 6.9282 I}}

وفي برنامج ماتيماتيكا يمكن استخدام الأمر Solve لحل مجموعة من المعادلات في وقت واحد كالاتي :

الصيغة العامة للأمر	العمل الذي يقوم به الأمر
Solve[eqns]	حل مجموعة المعادلات eqns بالنسبة الى جميع المتغيرات الموجودة فيها حيث eqns تكتب في صورة قائمة {lhs1==rhs1,lhs2==rhs2,...}
Solve[eqns,vars]	حل مجموعة المعادلات eqns بالنسبة الى المتغيرات vars حيث تكتب في صورة قائمة {x1,x2,...}
Solve[eqns, vars, elims]	حل مجموعة المعادلات eqns بالنسبة الى المتغيرات vars فقط وحذف المتغيرات elims من النتائج

In[20]:=Solve[{x+y==1,x-3y==2}] حل المعادلتين بالنسبة الى جميع المتغيرات  
Out[20]= {{x ->  $\frac{5}{4}$ , y ->  $-\frac{1}{4}$ } الموجودة وهي x, y

In[21]:=Solve[{x^2+y^2==5,x+3y==1}]/N للحصول على قيمة عددية لحل  
Out[21]={{x -> -2., y -> 1.}, {x -> 2.2, y -> -0.4}} جميع المتغيرات الى النسبة الى جميع المعادلتين

تعريف القائمة eqns1 وتحتوى على معادلتين فى ثلاث مجاهيل هم  $x, y, z$  ثم حساب  
الحل بالنسبة الى  $x, y$  فقط

In[22]:=eqns1={y-2x==9,x+3z==1}; Solve[eqns1,{x,y}]  
Out[22]={{y -> 9 - 2 (-1 + 3 z), x -> 1 - 3 z}}

حساب حل مجموعة المعادلات eqns1 بالنسبة الى  $y, z$  فقط

In[23]:=Solve[eqns1,{y,z}]

Out[23]={{y -> 9 + 2 x, z ->  $\frac{1-x}{3}$ }}

تعريف القائمة eqns2 وتحتوى على ثلاث معادلات فى خمسة مجاهيل هم  $x, y, z, w, t$   
ثم حساب الحل بالنسبة الى المتغيرات  $x, y, z$  فقط

In[24]:=eqns2={x+2y==z,y+2z==w,z+2w==t};  
Solve[eqns2,{x,y,z}]

Out[24]={{x -> t - 2 w + 2 (-w - 2 (-t + 2 w)),  
y -> w + 2 (-t + 2 w), z -> t - 2 w}}

In[25]:=Solve[eqns1,{x},{z}]

حساب حل مجموعة المعادلات eqns1

Out[25]={{x ->  $\frac{-9+y}{2}$ }}

بالنسبة الى المتغير  $x$  مع حذف المتغير  $z$

In[26]:=Solve[eqns1,{y},{z}]

حساب حل مجموعة eqns1

Out[26]={{y -> 9 + 2 x}}

بالنسبة إلى المتغير  $y$  مع حذف المتغير  $z$

In[27]:=Solve[eqns2,{x},{t,w}]

حساب حل مجموعة المعادلات eqns2

Out[27]={{x -> -2 y + z}}

بالنسبة إلى المتغير x مع حذف المتغيرات t,w

In[28]:=Solve[eqns2,{w},{x,y}]

حساب حل مجموعة المعادلات eqns2

Out[28]= {{w ->  $\frac{t-z}{2}$ }}

بالنسبة إلى المتغير w مع حذف المتغيرات x,y

وفي برنامج ماتيماتيكا يمكن حذف عدد من المتغيرات من مجموعة المعادلات وإعادة كتابتها ويتم ذلك باستخدام الأمر Eliminate كالتالي :

**Eliminate[eqns,elims]**

لحذف المتغيرات **elims** من مجموعة المعادلات **eqns**

In[29]:=Eliminate[eqns1,x]

لحذف المتغير x من مجموعة المعادلات eqns1

Out[29]=y == 11 - 6 z

In[30]:=Eliminate[eqns1,z]

لحذف المتغير x من مجموعة المعادلات eqns1

Out[30]=y == 9 + 2 x

In[31]:= Eliminate[eqns2,{x,y}]

لحذف المتغيرات x,y من مجموعة المعادلات eqns2

Out[31]=t == 2 w + z

In[32]:=Eliminate[eqns2,{w,t}]

لحذف المتغيرات w,t من مجموعة المعادلات eqns2

Out[32]=x == -2 y + z

## ٤ . الجبر الخطى Linear Algebra

يعتبر الجبر الخطى جزء اساسى وهام فى دراسة الرياضيات والهندسة والفيزياء وعلوم أخرى ، وبرنامج ماتيماتكا يقدم العديد من الأوامر للعمليات الجبرية الخاصة بالتعامل مع القوائم Lists والمصفوفات Matrices وحلول الأنظمة الخطية Linear Systems وحساب القيم المميزة والمتجهات المميزة لمصفوفة .

### أولا : القوائم Lists

من خلال دراستنا للعديد من الأوامر فى ماتيماتكا مثل

### Sum , Product , Series , ...

نلاحظ أن نطاق العمل فى هذه الأوامر يكتب باستخدام الأقواس { } على صورة قائمة List وتستخدم القوائم فى ماتيماتكا بشكل كبير وبصفة خاصة عند عمل الحسابات عندما يكون هناك حاجة لتنظيم عدد كبير من القيم بغرض التعامل معها كوحدة واحدة ولذلك فإن ماتيماتكا غنى بالعمليات التى يمكن تنفيذها على القوائم ، ولكى نعرف على هذه العمليات نبدأ بتعريف قائمتين s1 , s2 كل قائمة تحتوى على خمسة عناصر

In[1]:=s1={ a , b , c , d , e }; s2={ 2 , 3 , 4 , 5 , 6};

وتنفيذ العمليات الحسابية من جمع أو طرح أو ضرب أو قسمة على قائمتين يتم على العناصر المتناظرة فى القائمتين بشرط أن يكون القائمتين بهما نفس العدد من العناصر ونتائج العملية الحسابية يكون قائمة جديدة .

In[2]:= s1+s2

جمع القائمتين s1 , s2 يتم بجمع

Out[2]={2 + a , 3 + b , 4 + c , 5 + d , 6 + e}

العناصر المتناظرة فى القائمتين

$$\text{In}[3]:=s1-s2$$

طرح القائمتين  $s1, s2$  يتم بطرح

$$\text{Out}[3]=\{-2+a, -3+b, -4+c, -5+d, -6+e\}$$

العناصر المتناظرة في القائمتين

$$\text{In}[4]:=s1\ s2$$

ضرب القائمتين  $s1, s2$  يتم بضرب

$$\text{Out}[4]=\{2\ a, 3\ b, 4\ c, 5\ d, 6\ e\}$$

العناصر المتناظرة في القائمتين

$$\text{In}[5]:=s1/s2$$

قسمة القائمتين  $s1, s2$  يتم بقسمة

$$\text{Out}[5]=\left\{\frac{a}{2}, \frac{b}{3}, \frac{c}{4}, \frac{d}{5}, \frac{e}{6}\right\}$$

العناصر المتناظرة في القائمتين

$$\text{In}[6]:=s1+2$$

ويمكن إجراء أي عملية حسابية بين قائمة

$$\text{Out}[6]:\{a+2, b+2, c+2, d+2, e+2\}$$

وعدد ثابت فمثلا جمع القائمة  $s1$  على العدد

الثابت 2 يتم بإضافة العدد الثابت 2 الى

كل عنصر في القائمة

$$\text{In}[7]:=3s1$$

ضرب القائمة  $s1$  في العدد الثابت 3 يتم بضرب

$$\text{Out}[7]=\{3\ a, 3\ b, 3\ c, 3\ d, 3\ e\}$$

العدد الثابت 3 في كل عنصر من القائمة

$$\text{In}[8]:=s2^2$$

ويمكن رفع القائمة الى أس عددي حيث يتم رفع

$$\text{Out}[8]=\{4,9,16,25,36\}$$

كل عنصر في القائمة الى هذا الأس العددي

In[9]:= 2^s1

ويمكن رفع أي قيمة عددية الى أس عبارة عن قائمة

Out[9]= { 2<sup>a</sup> , 2<sup>b</sup> , 2<sup>c</sup> , 2<sup>d</sup> , 2<sup>e</sup> }

In[10]:=s1^s2

ويمكن رفع قائمة الى أس عبارة عن قائمة أخرى

Out[10]= { a<sup>2</sup> , b<sup>3</sup> , c<sup>4</sup> , d<sup>5</sup> , e<sup>6</sup> }

حيث يتم رفع كل عنصر في القائمة الأساس الى

أس يساوى العنصر المناظر له في القائمة الأس

ويمكن تطبيق الدوال على القوائم حيث يتم تطبيق الدالة على كل عنصر في القائمة

إيجاد الجذر التربيعي للقائمة s2 حيث يتم حساب الجذر التربيعي لكل عنصر في القائمة

In[11]:=Sqrt[s2]/N

Out[11]= {1.41421, 1.73205, 2., 2.23607, 2.44949}

تطبيق دالة الجيب sin على القائمة s2

In[12]:=Sin[s2]/N

Out[12]={0.909297, 0.14112, -0.756802, -0.958924, -0.279415}

وماتيماتكا قادر على إجراء عمليات الفئات Sets على القوائم من خلال العديد من الأوامر والجدول الآتي يوضح بعض الأوامر المستخدمة .

الصيغة العامة للأمر	العمل الذي يقوم به الأمر
Length[list]	إيجاد عدد العناصر في القائمة list
Sort[list]	ترتيب عناصر القائمة list حيث يتم أولاً ترتيب الأعداد تصاعدياً ثم ترتيب الحروف أبجدياً
Join[list1, list2, ...]	إضافة القوائم list1 , list2, ... على بعضها بحيث تحتوي القائمة الناتجة على عدد من العناصر يساوي مجموع أعداد العناصر في كل قائمة
Union[list1, list2, ...]	اتحاد الفئات $list1 \cup list2 \cup list3 \cup \dots$ حيث يتم حذف العناصر المكررة في القوائم
Intersection[list1, list2, ...]	تقاطع الفئات $list1 \cap list2 \cap list3 \cap \dots$
Complement[eall, e1, e2, ...]	إيجاد مكملة الفئة eall بالنسبة للفئات e1 , e2 , ... أي إيجاد العناصر في الفئة eall والغير موجودة في الفئات e1 , e2 , ...
Partition[list, n]	تجزئ القائمة list الى قوائم فرعية متباعدة كل منها يحتوي على n من العناصر


ولتوضيح أوامر الفئات نفرض القوائم  $X, Y, Z$

In[13]:=X={a,b,c,d,e};Y={b,d,v};Z={a,b,e,u,w};

In[14]:=Length[X]

لمعرفة عدد العناصر في X

Out[14]=5



اضافة القوائم  $X, Y, Z$  معا بحيث تحتوي القائمة الناتجة على عدد من العناصر يساوى مجموع أعداد العناصر في كل قائمة


In[15]:=Join[X,Y,Z]

Out[15]={a, b, c, d, e, b, d, v, a, b, e, u, w}

اتحاد الفئات  $XUYUZ$  حيث يتم حذف العناصر المكررة في القوائم

In[16]:=Union[X,Y,Z]


Out[16]={a, b, c, d, e, u, v, w}



تقاطع الفتان  $X \cap Y$  وتمثل فئة العناصر المشتركة في الفتان  $X, Y$

In[17]:=Intersection[X,Y]


Out[17]={b, d}



تقاطع الفتان  $X \cap Z$  وتمثل فئة العناصر المشتركة في الفتان  $X, Z$

In[18]:=Intersection[X,Z]

Out[18]={a, b, e}



إيجاد مكملة الفئة X بالنسبة للفئة Y أي إيجاد العناصر في الفئة X والغير موجودة في الفئة Y

In[19]:=Complement[X,Y]

Out[19]={a, c, e}

In[20]:=Partition[Z,2]

Out[20]={{a, b}, {e, u}}

تجزئ الفئة Z الى قوائم فرعية كل

منها يحتوى على عنصرين

In[21]:=Partition[Y,1]

Out[21]={{b}, {d}, {v}}

تجزئ الفئة Y الى قوائم فرعية كل

منها يحتوى على عنصر واحد فقط

In[22]:=Sort[{r,4,9,p,e,a,-7}]

Out[22]={-7, 4, 9, a, e, p, r}

ترتيب عناصر القائمة حيث يتم أولا

ترتيب الأعداد تصاعديا ثم ترتيب

الحروف أبجديا

ويمكن إضافة عناصر جديدة الى القوائم باستخدام الأوامر الآتية :

الصيغة العامة للأمر	العمل الذى يقوم به الأمر
<b>Prepend[list, elem]</b>	إضافة العنصر <b>elem</b> فى بداية القائمة <b>list</b>
<b>Append[list, elem]</b>	إضافة العنصر <b>elem</b> فى نهاية القائمة <b>list</b>
<b>Insert[list, elem, n]</b>	إضافة العنصر <b>elem</b> الى القائمة <b>list</b> فى الموضع رقم <b>n</b>

**In[23]:=rrr={a,b,c,d,e};**

تعريف القائمة **rrr**

**In[24]:=Prepend[rrr,x]**

إضافة العنصر **x** فى بداية القائمة **rrr**

**Out[24]={x, a, b, c, d, e}**

**In[25]:=Append[rrr,y]**

إضافة العنصر **y** إلى نهاية القائمة **rrr**

**Out[25]={a, b, c, d, e, y}**

**In[26]:=Insert[rrr,z,2]**

إدخال العنصر **z** فى الموضع رقم 2 من القائمة **rrr**

**Out[26]={a, z, b, c, d, e}**

ويمكن حذف عناصر من القوائم باستخدام الأمر Drop كالاتي :

الصيغة العامة للأمر	العمل الذي يقوم به الأمر
<b>Drop[list, n]</b>	حذف n من العناصر من بداية القائمة list
<b>Drop[list, -n]</b>	حذف n من العناصر من نهاية القائمة list
<b>Drop[list,{n}]</b>	حذف العنصر رقم n من القائمة list
<b>Drop[list, {m, n}]</b>	حذف عناصر من القائمة list ابتداء من العنصر رقم m الى العنصر رقم n

**In[27]:=r1={a1,a2,a3,a4,a5,a6,a7};**

حذف ثلاثة عناصر من بداية القائمة r1

**Drop[r1,3]**

**Out[27]={a4, a5, a6, a7}**

**In[28]:=Drop[r1,-2]**

حذف عنصران من نهاية القائمة r1

**Out[28]={a1, a2, a3, a4, a5}**

**In[29]:=Drop[r1,{4}]**

حذف العنصر الرابع من القائمة r1

**Out[29]={a1, a2, a3, a5, a6, a7}**

**In[30]:=Drop[r1,{3,6}]**

حذف العناصر من العنصر الثالث الى

**Out[30]={a1, a2, a7}**

العنصر السادس من القائمة r1

ويمكن تحديد عناصر معينة من القائمة وذلك باستخدام الأقواس المزدوجة `[[ ]]`

`In[31]:=r1[[4]]`

لتحديد العنصر الرابع من القائمة `r1`

`Out[31]=a4`

`In[32]:=r1[[{4,6}]]`

ولتحديد العنصران الرابع والسادس من القائمة `r1`

`Out[32]={a4, a6}`

وفي برنامج ماتيماتكا يمكن توليد قوائم بناء على مواصفات نحددها له وذلك باستخدام الأمر `Table` كالتالي :

الصيغة العامة للأمر	العمل الذي يقوم به الأمر
<code>Table[expr, {imax}]</code>	عمل قائمة تحتوي على نسخ من <code>expr</code> عددها <code>imax</code>
<code>Table[expr, {i, imax}]</code>	عمل قائمة تحتوي على قيم <code>expr</code> ابتداء من <code>i=1</code> حتى <code>i=imax</code>
<code>Table[expr, {i, imin, imax}]</code>	عمل قائمة تحتوي على قيم <code>expr</code> ابتداء من <code>i=imin</code> حتى <code>i=imax</code>
<code>Table[expr, {i, imin, imax, di}]</code>	عمل قائمة تحتوي على قيم <code>expr</code> ابتداء من <code>i=imin</code> حتى <code>i=imax</code> بخطوة <code>di</code> مقدارها <code>di</code>
<code>Table[expr, {i, imin, imax}, {j, jmin, jmax}, ...]</code>	عمل جدول من القوائم يحتوي على قيم <code>expr</code> في أكثر من بعد <code>i, j, ...</code>
<code>TableForm[list]</code>	كتابة القائمة <code>list</code> في الشكل التقليدي للمصفوفة

ولتوضيح عمل الأمر *Table* نعطي الأمثلة الآتية :

In[33]:=Table[x,{4}]

لتوليد قائمة تحتوي على أربعة نسخ من الرمز x

Out[33]= {x, x, x, x}

In[34]:=Table[Random[],5]

لتوليد قائمة تحتوي على خمسة أعداد

Out[34]=

عشوائية في الفترة [0,1]

{0.803812, 0.152706, 0.0624843, 0.59723, 0.192153}

In[35]:=Table[i^2,{i,7}]

لتوليد قائمة تحتوي على قيم  $i^2$  من

Out[35]= {1, 4, 9, 16, 25, 36, 49}

$i=1$  الى  $i=7$

In[36]:=Table[x^i + 2i x, {i,3,6}]

لتوليد قائمة تحتوي على المقدار الجبري

Out[36]=

$x^i + 2ix$  من  $i=3$  الى  $i=6$

{6 x + x<sup>3</sup>, 8 x + x<sup>4</sup>, 10 x + x<sup>5</sup>, 12 x + x<sup>6</sup>}

In[37]:=Table[i^3, {i,2,8,2}]

لتوليد قائمة عناصرها هي مكعبات الأعداد

Out[37]= {8, 64, 216, 512}

الزوجية المحصورة بين 2, 8

In[38]:=m=Table[i^2+2j, {i,3}, {j,2,5}]

لتوليد قائمة m تحتوي على قيم  $i^2 + 2j$

Out[38]:=

حيث  $i=1,2,3$  &  $j=2,3,4,5$

{{5, 7, 9, 11}, {8, 10, 12, 14}, {13, 15, 17, 19}}

In[39]:= TableForm[m]

ولعرض القائمة m في صورة جدول

Out[39]=

5	7	9	11
8	10	12	14
13	15	17	19

ثانيا : المصفوفات Matrices

المصفوفات والعمليات الخاصة بها تستخدم بشكل كبير في الرياضيات ويمكن الاستفادة من ماتيماتكا في إجراء العمليات الرياضية الخاصة بالمصفوفات والتي كانت تستغرق الكثير من الوقت خاصة إذا كانت المصفوفات من أبعاد كبيرة . والمصفوفات في ماتيماتكا عبارة عن قوائم من قوائم lists of lists فمثلا

- القائمة {a,b,c} تمثل المتجه (a,b,c) وهي مصفوفة من صف واحد وثلاثة اعمدة
- والقائمة {{a,b},{c,d}} تمثل المصفوفة  $\begin{pmatrix} a & b \\ c & d \end{pmatrix}$  وهي مصفوفة من صفين وعمودين
- والقائمة {{a<sub>1</sub>,a<sub>2</sub>,a<sub>3</sub>},{b<sub>1</sub>,b<sub>2</sub>,b<sub>3</sub>}} تمثل المصفوفة  $\begin{pmatrix} a_1 & a_2 & a_3 \\ b_1 & b_2 & b_3 \end{pmatrix}$  وهي مصفوفة من صفين وثلاثة اعمدة وهكذا ،

وبالتالى فانه يمكن إدخال عناصر المصفوفة بصورة القوائم ، ويحتوى ماتيماتكا على الأوامر Table , Array والخاصة بتكوين مصفوفات ذات أبعاد مختلفة والأمر Table هو الأكثر استخداما .

الصيغة العامة للأمر	العمل الذى يقوم به الأمر
Table[f,{i,m},{j,n}]	تكوين مصفوفة من البعد mxn حيث m تمثل عدد الصفوف ، n تمثل عدد الأعمدة ، f تمثل دالة فى i,j لتوليد عناصر المصفوفة
Array[f, n]	تكوين مصفوفة على شكل صف به n من العناصر على الصورة f[i]
Array[f, {m,n}]	تكوين مصفوفة من البعد mxn على الصورة {f[i,j]} حيث [i,j] يمثل العنصر فى الصف i والعمود j
MatrixForm[list]	طباعة القائمة list فى الشكل التقليدى للمصفوفة

وفى ماتيماتكا يمكن اجراء العمليات الحسابية من جمع وطرح وضرب على المصفوفات وتوضيح ذلك

In[1]:= Array[h,6]                      تكوين مصفوفة على شكل صف به 6 عناصر  
Out[1]={h[1], h[2], h[3], h[4], h[5], h[6]}

In[2]:=Array[a,{2,2}]                      تكوين مصفوفة 2x2 عناصرها على صورة  $a_{ij}$   
Out[2]={{a[1, 1], a[1, 2]}, {a[2, 1], a[2, 2]}}

تكوين مصفوفة 3x3 عناصرها على الصورة  $f_{ij} = 10i + j$  ثم عرض الناتج  
فى الشكل التقليدى للمصفوفة

In[3]:=f=Table[10i+j,{i,3},{j,3}]  
;MatrixForm[f]

Out[3]= 
$$\begin{matrix} 11 & 12 & 13 \\ 21 & 22 & 23 \\ 31 & 32 & 33 \end{matrix}$$

تكوين مصفوفة 3x3 عناصرها على صورة  $m_{ij}$

In[4]:=g=Array[m,{3,3}]

Out[4]={{m[1, 1], m[1, 2], m[1, 3]},  
{m[2, 1], m[2, 2], m[2, 3]},  
{m[3, 1], m[3, 2], m[3, 3]}}

حساب مجموع المصفوفتان  $f$  ,  $g$  ثم عرض الناتج في الشكل التقليدي للمصفوفة

**In[5]:=MatrixForm[f+g]**

**Out[5]=**  $11 + m[1, 1]$        $12 + m[1, 2]$        $13 + m[1, 3]$   
 $21 + m[2, 1]$        $22 + m[2, 2]$        $23 + m[2, 3]$   
 $31 + m[3, 1]$        $32 + m[3, 2]$        $33 + m[3, 3]$

حساب حاصل طرح المصفوفتان  $f$  ,  $g$  ثم عرض الناتج في الشكل التقليدي للمصفوفة

**In[6]:=MatrixForm[f-g]**

**Out[6]=**  $11 - m[1, 1]$        $12 - m[1, 2]$        $13 - m[1, 3]$   
 $21 - m[2, 1]$        $22 - m[2, 2]$        $23 - m[2, 3]$   
 $31 - m[3, 1]$        $32 - m[3, 2]$        $33 - m[3, 3]$

حساب حاصل ضرب المصفوفة  $g$  في العدد 5

**In[7]:=5g**

**Out[7]=** {  $5 m[1, 1]$  ,  $5 m[1, 2]$  ,  $5 m[1, 3]$  } ,  
{  $5 m[2, 1]$  ,  $5 m[2, 2]$  ,  $5 m[2, 3]$  } ,  
{  $5 m[3, 1]$  ,  $5 m[3, 2]$  ,  $5 m[3, 3]$  } }

حساب خارج قسمة المصفوفة  $g$  على  $3$

In[8]:=  $g/3$

Out[8]=

$$\left\{ \left\{ \frac{m(1,1)}{3}, \frac{m(1,2)}{3}, \frac{m(1,3)}{3} \right\}, \left\{ \frac{m(2,1)}{3}, \frac{m(2,2)}{3}, \frac{m(2,3)}{3} \right\}, \left\{ \frac{m(3,1)}{3}, \frac{m(3,2)}{3}, \frac{m(3,3)}{3} \right\} \right\}$$

حساب حاصل ضرب المصفوفتان  $f$  ,  $g$

In[9]:=  $f . g$

$$\begin{aligned} \text{Out[9]=} & \{ \{ 11 m[1, 1] + 12 m[2, 1] + 13 m[3, 1], \\ & 11 m[1, 2] + 12 m[2, 2] + 13 m[3, 2], \\ & 11 m[1, 3] + 12 m[2, 3] + 13 m[3, 3] \}, \\ & \{ 21 m[1, 1] + 22 m[2, 1] + 23 m[3, 1], \\ & 21 m[1, 2] + 22 m[2, 2] + 23 m[3, 2], \\ & 21 m[1, 3] + 22 m[2, 3] + 23 m[3, 3] \}, \\ & \{ 31 m[1, 1] + 32 m[2, 1] + 33 m[3, 1], \\ & 31 m[1, 2] + 32 m[2, 2] + 33 m[3, 2], \\ & 31 m[1, 3] + 32 m[2, 3] + 33 m[3, 3] \} \} \end{aligned}$$

تعريف المصفوفة  $f1$  من رتبة  $3 \times 2$  وتعريف المصفوفة  $f2$  من رتبة  $2 \times 4$  ثم حساب

حاصل ضرب المصفوفتان  $f1$  ,  $f2$  وعرض الناتج في الشكل التقليدي للمصفوفة

In[10]:=  $f1 = \{\{2,-1\},\{1,0\},\{-3,4\}\}; f2 = \{\{1,-2,3,0\},\{3,4,0,1\}\};$   
 $\text{MatrixForm}[f1.f2]$

Out[10]=

-1	-8	6	-1
1	-2	3	0
9	22	-9	4

وفي ماتيماتيكا يمكن إجراء العمليات الأساسية على المصفوفات كالآتي :

الصيغة العامة للأمر	العمل الذي يقوم به الأمر
<b>Transpose[m]</b>	حساب مدور المصفوفة m
<b>Det[m]</b>	حساب قيمة محدد المصفوفة m
<b>Minors[m, k]</b>	حساب مصفوفة المحددات المصاحبة من رتبة kxk من المصفوفة m
<b>Inverse[m]</b>	حساب معكوس المصفوفة المربعة m
<b>MatrixPower[m,k]</b>	حساب $m^k$

**In[11]:=g1=Transpose[g];MatrixForm[g1]** حساب مدور المصفوفة g ثم  
**Out[11]=**  $\begin{matrix} m[1, 1] & m[2, 1] & m[3, 1] \\ m[1, 2] & m[2, 2] & m[3, 2] \\ m[1, 3] & m[2, 3] & m[3, 3] \end{matrix}$  عرض الناتج في الشكل التقليدي للمصفوفة

**In[12]:=g2={{2,4},{1,7}};Det[g2]** تعريف المصفوفة g2 من رتبة 2x2  
**Out[12]=10** ثم حساب قيمة المحدد

In[13]:=Inverse[g2]

حساب معكوس المصفوفة g2

Out[13]= $\left\{\left\{\frac{7}{10}, -\left(\frac{2}{5}\right)\right\}, \left\{-\left(\frac{1}{10}\right), \frac{1}{5}\right\}\right\}$

In[14]:=MatrixPower[g2,3]

لحساب  $(g2)^3$

Out[14]= $\left\{\{52, 284\}, \{71, 407\}\right\}$

تعريف المصفوفة g3 من رتبة 3x3 ثم حساب مصفوفة المحددات المصاحبة

من رتبة 2x2 من المصفوفة g3

In[15]:=g3={{1,5,7},{2,4,3},{-1,6,0}}; Minors[g3,2]

Out[15]= $\left\{\{-6, -11, -13\}, \{11, 7, -42\}, \{16, 3, -18\}\right\}$

In[16]:=d=Det[g3] }

حساب قيمة المحدد للمصفوفة g3

Out[16]=279

In[17]:=Inverse[g3]

حساب معكوس المصفوفة g3

Out[17]=

$\left\{\left\{-\left(\frac{18}{79}\right), \frac{42}{79}, -\left(\frac{13}{79}\right)\right\}, \left\{-\left(\frac{3}{79}\right), \frac{7}{79}, \frac{11}{79}\right\}, \left\{\frac{16}{79}, -\left(\frac{11}{79}\right), -\left(\frac{6}{79}\right)\right\}\right\}$

$f[i]$	لتحديد الصف رقم $i$ في المصفوفة $f$
$f[i,j]$	لتحديد العنصر $f[i,j]$ بالصف رقم $i$ والعمود رقم $j$ في المصفوفة $f$
$\text{Sum}[f[[i,i]],\{i,n\}]$	لحساب مجموع عناصر القطر الرئيسي في المصفوفة المربعة $f$ من رتبة $n \times n$
$\text{Transpose}[f][[j]]$	لتحديد العمود رقم $j$ في المصفوفة $f$

In[18]:=g[[2]]

تحديد الصف الثاني من المصفوفة  $g$

Out[18]={m[2, 1], m[2, 2], m[2, 3]}

In[19]:=g[[3,1]]

لتحديد العنصر الموجود في الصف

Out[19]=m[3, 1]

الثالث والعمود الأول في المصفوفة  $g$

In[20]:=Sum[g[[i,i]],\{i,3\}]

لحساب مجموع عناصر القطر الرئيسي

Out[20]=m[1, 1] + m[2, 2] + m[3, 3]

في المصفوفة  $g$

In[21]:=Transpose[g][[3]]

لتحديد العمود الثالث من المصفوفة g

Out[21]={m[1, 3], m[2, 3], m[3, 3]}

ويستطيع برنامج ماتيماتيكا تكوين مصفوفات من اشكال مختلفة كالاتي :

الصيغة العامة للأمر	العمل الذي يقوم به الأمر
DiagonalMatrix[list]	تكوين مصفوفة قطرية بحيث أن عناصر القائمة list توضع في قطر المصفوفة وباقي العناصر أصفار
IdentityMatrix[n]	تكوين مصفوفة الوحدة من البعد nxn
Table[0,{m},{n}]	تكوين مصفوفة صفرية من البعد mxn
Table[If[i<=j,1,0],{i,m},{j,n}]	تكوين مصفوفة مثلثية عليا Upper Triangular عناصرها في أعلى القطر 1 وخلاف ذلك أصفار

In[22]:=DiagonalMatrix[{a,b,c}]

تكوين مصفوفة قطرية من القائمة {a,b,c}

Out[22]={{a, 0, 0}, {0, b, 0}, {0, 0, c}}

In[23]:=IdentityMatrix[3]

تكوين مصفوفة الوحدة من البعد 3x3

Out[23]={{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}

In[24]:=Table[0,{i,3},{j,2}]

تكوين مصفوفة صفرية من البعد 3x2

Out[24]={{0,0},{0,0},{0,0}}

In[25]:=Table[If[i<=j,1,0],{i,3},{j,3}]

تكوين مصفوفة مثلثية عليا من البعد 3x3

Out[25]={{1, 1, 1}, {0, 1, 1}, {0, 0, 1}}

In[26]:=Table[If[i>=j,1,0],{i,4},{j,4}]

تكوين مصفوفة مثلثية سفلى من البعد 4x4

Out[26]=

{{1, 0, 0, 0}, {1, 1, 0, 0}, {1, 1, 1, 0}, {1, 1, 1, 1}}

In[27]:=MatrixForm[%26]

لعرض المصفوفة الناتجة من جملة الإدخال

Out[27]= 1 0 0 0

في In[26] الشكل التقليدي للمصفوفة

1 1 0 0

1 1 1 0

1 1 1 1

## ثالثا : حل الأنظمة الخطية Solving Linear Systems

نظرية المعادلات الخطية linear equations تلعب دورا هاما في الجبر الخطي , وفى الحقيقة فإن دراسة مسائل عديدة فى الجبر الخطي يتم تحويلها الى دراسة نظام من المعادلات الخطية . والمعادلة الخطية هى معادلة لها الصورة  $a_1 x_1 + a_2 x_2 + \dots + a_n x_n = b$  حيث  $x_i$  تمثل متغيرات ,  $a_i$  أعداد حقيقية وتمثل معاملات المتغيرات ,  $b$  عدد حقيقى ويمثل الحد الثابت .

وكثيرا ما نحتاج الى إيجاد حلول أنظمة من المعادلات الخطية , وفى بعض الحالات يكون من الأفضل كتابة المعادلات ثم حلها باستخدام الأمر Solve وفى حالات أخرى يكون من المفيد تحويل نظام المعادلات الخطية الى شكل مصفوفات  $m \cdot x = b$  حيث  $x$  يمثل متجه المتغيرات ,  $m$  يمثل مصفوفة المعاملات ,  $b$  يمثل متجه الثوابت .

حل المعادلتين باستخدام الأمر Solve مباشرة `In[1]:=Solve[{x+5y==1,2x+y==3},{x,y}]`

`Out[1]={{x->14/9,y->-1/9}}`

تعريف المصفوفة `mat1` من رتبة  $3 \times 3$  `In[2]:=mat1={{2,1,-2},{3,2,2},{5,4,3}}`

`Out[2]={{2, 1, -2}, {3, 2, 2}, {5, 4, 3}}`

تكوين نظام من ثلاث معادلات `In[3]:=mat1.{x,y,z}=={10,1,4}`

`Out[3]={2 x + y - 2 z, 3 x + 2 y + 2 z,  
5 x + 4 y + 3 z} == {10, 1, 4}`

حل نظام المعادلات باستخدام الأمر `In[4]:=Solve[%,{x,y,z}]`

`Out[4]={{x -> 1, y -> 2, z -> -3}}` Solve مباشرة

حل نظام المعادلات باستخدام معكوس المصفوفة `In[5]:= {x,y,z}=Inverse[mat1].{10,1,4}`

`Out[5]={1, 2, -3}`

وبرنامج ماتيماتيكا قادر على حل نظام المعادلات الخطية في صورة مصفوفة باستخدام الأمر **LinearSolve** كالآتي :

<b>LinearSolve[m, b]</b>	لايجاد متجه المتغيرات $x$ الذي يحقق نظام المعادلات الخطية $m \cdot x = b$
--------------------------	--

وعند التعامل مع مصفوفات من ابعاد كبيرة يكون من الافضل استخدام الأمر **LinearSolve** لحل نظام المعادلات .

حل نظام المعادلات الخطية

$$2x + y - 2z = 10$$

$$3x + 2y + 2z = 1$$

$$5x + 4y + 3z = 4$$

في صورة مصفوفة باستخدام الأمر **LinearSolve** يكون كالآتي:

**In[6]:=LinearSolve[mat1,{10,1,4}]**

**Out[6]={1, 2, -3}**

حيث **mat1** هي مصفوفة المعاملات  $\begin{pmatrix} 2 & 1 & -2 \\ 3 & 2 & 2 \\ 5 & 4 & 3 \end{pmatrix}$  في نظام المعادلات المعرف في جملة

الإدخال **In[3]** والحل يكون

$$x = 1 , y = 2 , z = -3$$

حل نظام المعادلات الخطية

$$\begin{pmatrix} 1 & 2 & -3 \\ 2 & -1 & 4 \\ 4 & 3 & -2 \end{pmatrix} \begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} 6 \\ 2 \\ 14 \end{pmatrix}$$

في صورة مصفوفة باستخدام الأمر **LinearSolve** يكون كالآتي :

**In[7]:=mat2={{1,2,-3},{2,-1,4},{4,3,-2}};**

**LinearSolve[mat2,{6,2,14}]**

**Out[7]={2, 2, 0}**

$$x = 2 , y = 2 , z = 0$$

والحل يكون

حل نظام المعادلات الخطية

$$\begin{pmatrix} 1 & -3 & 4 & -2 \\ 0 & 2 & 5 & 1 \\ 0 & 1 & -3 & 0 \end{pmatrix} \begin{pmatrix} x \\ y \\ z \\ w \end{pmatrix} = \begin{pmatrix} 5 \\ 2 \\ 4 \end{pmatrix}$$

في صورة مصفوفة باستخدام الأمر **LinearSolve** يكون كالآتي :

**In[8]:=mat3={{1,-3,4,-2},{0,2,5,1},{0,1,-3,0}};**

**LinearSolve[mat3,{5,2,4}]**

**Out[8]={ $\frac{157}{11}, \frac{26}{11}, -(\frac{6}{11}), 0$ }**

ونلاحظ أن نظام المعادلات يتكون من ثلاثة معادلات في أربعة مجاهيل وله عدد لا نهائي من الحلول لأنه تم حساب ثلاث مجاهيل بدلالة المجهول الرابع ونتائج الحل يمثل حل نظام المعادلات بعد اخذ قيمة عددية للمجهول الرابع .

وإذا كان  $m$  مصفوفة مربعة فإننا نعلم من دراستنا في الجبر الخطي أن نظام المعادلات

$$m \cdot x = b$$

يكون له حل وحيد لأي متجه ثوابت  $b$  إذا كان محدد المصفوفة  $m$  لا يساوى صفر بينما إذا كان محدد المصفوفة  $m$  يساوى صفر فهذا يعني انه لا يوجد حل  $x$  يحقق نظام المعادلات  $m \cdot x = b$  . لقيمة خاصة للمتجه  $b$  وبمعنى آخر أن المعادلات تعتمد على بعضها  $dependent$  وفي هذه الحالة فإن مجموعة المتجهات  $x$  التي تحقق  $m \cdot x = 0$  تسمى الفراغ الصفري  $nullspace$  للمصفوفة  $m$  أو  $kernel m$  وماتيماتكا قادر على حساب متجهات الأساس  $Basis$  للفراغ الصفري لمصفوفة وذلك باستخدام الأمر  $NullSpace$  كالآتي :

### NullSpace[m]

لإيجاد متجهات أساس  $basis\ vectors$  جميع تركيباتها الخطية  $linear$  combinations تحقق المعادلة  $m \cdot x = 0$  حيث  $0$  هو المتجه الصفري

In[9]:=mat4={{1,2,1},{2,4,2},{3,6,3}} محدد المصفوفة mat4 يساوى صفر

;Det[mat4]

Out[9]=0

In[10]:=LinearSolve[mat4,{a,b,c}] الدالة LinearSolve لا تستطيع إيجاد

Out[10]=LinearSolve::nosol: حل نظام المعادلات

Linear equation encountered which has no solution.

LinearSolve[{{1, 2, 1}, {2, 4, 2}, {3, 6, 3}}, {a, b, c}]

In[11]:=NullSpace[mat4] أساس الفراغ الصفري للمصفوفة mat4

Out[11]={{-1, 0, 1}, {-2, 1, 0}} يحتوى على متجهان

ومن المميزات الهامة للأمر LinearSolve والأمر NullSpace هو التعامل مع مصفوفات من أي رتبة .

رابعاً : القيم المميزة والمتجهات المميزة Eigenvalues and Eigenvectors

القيم المميزة لمصفوفة  $m$  هي قيم  $\lambda$  التي تحقق المعادلة  $m \cdot \lambda \cdot x$  حيث  $x$  متجه غير صفري وفي هذه الحالة فإن المتجهات  $x$  التي تحقق هذه المعادلة تسمى المتجهات المميزة للمصفوفة  $m$  والتي تناظر القيمة المميزة  $\lambda$  ويمكن الحصول على القيم المميزة من حل كثيرة الحدود المميزة **Characteristic polynomial** وتعطى بالمعادلة

$$|m - \lambda \cdot J| = 0$$

حيث  $J$  مصفوفة الوحدة . وبرنامج ماتيماتيكاً قادر على حساب القيم المميزة والمتجهات المميزة لأي مصفوفة كالاتي :

الصيغة العامة للأمر	الوظيفة التي يقوم بها الأمر
Eigenvalues[m]	تكوين قائمة تحتوى على جميع القيم المميزة للمصفوفة $m$
Eigenvectors[m]	تكوين قائمة تحتوى على جميع المتجهات المميزة للمصفوفة $m$
Eigensystem[m]	تكوين قائمة تحتوى على جميع القيم المميزة و المتجهات المميزة للمصفوفة $m$ وتكون بالصورة { eigenvalue,eigenvector }
Eigenvalues[N[m]]	تكوين قائمة تحتوى على قيم عددية تقريبية للقيم المميزة للمصفوفة $m$

In[1]:=m={{1,2},{3,2}} تعريف المصفوفة  $\begin{pmatrix} 1 & 2 \\ 3 & 2 \end{pmatrix}$

Out[1]= {{1, 2}, {3, 2}}

In[2]:=Eigenvalues[m]

Out[2]= Eigenvalues::eival:

Unable to find all roots of the characteristic polynomial.

Eigenvalues[{{1, 2}, {3, 2}}]

نلاحظ أن ماتيماتيكا لا يتمكن من حساب القيم المميزة للمصفوفة  $m$  لأن عناصر المصفوفة أعداد صحيحة ويمكن التغلب على هذه المشكلة عن طريق كتابة الأعداد الموجودة في المصفوفة بالصورة العشرية (فمثلا يكتب 3 بدلا من 3) كما يمكن عمل ذلك باستخدام الدالة  $N$

In[3]:=Eigenvalues[N[m]]

القيم المميزة للمصفوفة  $m$

Out[3]= {4., -1.}

In[4]:=Eigenvectors[N[m]]

المتجهات المميزة للمصفوفة  $m$

Out[4]= {{-0.5547, -0.83205}, {-0.707107, 0.707107}}

In[5]:={values,vectors}=Eigensystem[N[m]] تكوين قائمة تحتوى على جميع القيم

Out[5]= {{4., -1.}, {{-0.5547,-0.83205}, {-0.707107, 0.707107}}} المميزة و المتجهات المميزة للمصفوفة  $m$

للتحقق من أن القيمة المميزة الأولى  $k_1$  والمتجه المميز  $x$  يحقق المعادلة  $m \cdot x = k_1 x$

**In[6]:=m.vectors[[1]]==values[[1]] vectors[[1]]**

**Out[6]= True**

لإيجاد المعادلة المميزة للمصفوفة  $m$

**In[7]:=po=Det[m-k IdentityMatrix[Dimensions[m]][[1]]]**

**Out[7]= -4 - 3 k + k<sup>2</sup>**

حل المعادلة المميزة والحصول على القيم المميزة للمصفوفة  $m$

**In[8]:=Solve[po==0,k]**

**Out[8]= {{k -> -1}, {k -> 4}}**